arXiv:hep-ph/0607049 v1 5 Jul 2006

Excursions into FeynArts and FormCalc

T. Hahn^a, J.I. Illana^b

aMax-Planck-Institut für Physik, Föhringer Ring 6, D–80805 Munich, Germany

bDepartamento de Física Teórica y del Cosmos, and Centro Andaluz de Física de Partículas Elementales (CAFPE), Universidad de Granada, E–18071 Granada, Spain

Programming techniques which extend the capabilities of FeynArts and FormCalc are introduced and explained using examples from real applications.

1. Introduction

FeynArts [1] and FormCalc [2] are programs for the generation and calculation of Feynman diagrams. The main problem they solve is the computation of the cross-section for a given process, and indeed this is meanwhile a heavily automated procedure that requires little user input. Feynman diagrams have a lot of other uses, however, and this note shows some examples of how to extend the capabilities of FeynArts and Form- Calc to produce results not within the ‘standard canon.’

Although cash-strapped institutes may tend to believe otherwise, it is actually a feature, not a bug, that FeynArts and FormCalc are largely written in Mathematica, for it is precisely the availability of a powerful language that makes it easy for the user to examine and modify results, and also to extend the functionality of these systems. Mathematica is particularly suited for the latter as it is technically an expert system, where knowledge is added in the form of transformation rules. Thus one usually does not have to modify the existing program code (possibly including recompilation), but just add new rules. Sections 2, 3, 4, and 5 show some real-life examples of such extensions. Section 6 exhorts on the concept of abbreviations in FormCalc, where some important improvements have been made recently.

2. Programming Diagram Filters

Question: What if FeynArts’ diagram selection functions are not enough? What if I want, say, only diagrams with a fermion loop?

Answer: Inspect the internal structure of the inserted topologies, i.e. the InsertFields output. The outermost structure is a TopologyList:

TopologyList[__][t_1, t_2, ...]

Contained in this are topologies t_i of the form

Topology[_][__] ->
  Insertions[Generic][g_1, g_2, ...]

The generic insertions g_i specify the fields running on each line of the topology, e.g.

Graph[__][Field[1] -> F, ...] -> c_i

Now recall that FeynArts distinguishes three levels of fields (generic, classes, particles) where the field’s space-time properties are fixed at the lowest (generic) level [3].

The fermion-ness of a field is a space-time property, so we need not go deeper than the generic level. In order to look at generic-level diagrams, the selection function for fermion loops must first “propagate down” into the generic level:

FermionLoop[t:TopologyList[___][__]] :=
  FermionLoop/@ t

FermionLoop[(top:Topology[_][__]) ->
        ins:Insertions[Generic][__]] :=
  top -> TestLoops[top]/@ ins

The subsidiary function TestLoops selects only the fields carried by propagators in the loop:

TestLoops[top_] := TestFerm[ Cases[top,
  Propagator[_Loop][_, _, field_] ->
    field] ]

1 and hands on to TestFerm which tests if, for a concrete diagram, the selected fields are indeed all fermions:

TestFerm[fields_][gi_ -> ci_] :=
  (gi -> ci) /;
    MatchQ[fields /. List@@ gi, {F..}]

TestFerm[_][_] := Sequence[]

Note this typical (for Mathematica) construct: the first instance handles diagrams positively identified as having a fermion loop, and the second one is the ‘fall-through’ for all other cases.

3. Tweaking Model Files

Question: Is there a good way to make (small) changes to an existing model file?

Answer: It is bad practice to copy the model file and modify the copy, for two reasons:

1. It is typically not very transparent what has changed.

2. If the original model file changes (e.g. due to bug fixes), these do not automatically propagate into the derivative model file.

Better: Create a new model file which reads the old one and modifies the particles and coupling tables. To this end one needs to know that the model file defines two main objects:

• M$ClassesDescription is the list of particle definitions,

• M$CouplingMatrices is the list of couplings of the form C[fields] == expr .

As an example consider introducing enhancement factors for the b–b̄–h_0 and b–b̄–H_0 Yukawa couplings in the MSSM. The new model file is both compact and makes it immediately obvious what has changed:

Block[{$Path = $ModelPath}, << MSSM.mod]

EnhCoup[ (lhs:C[F[4,{g_,_}], -F[4,_],
                 S[h:1|2]]) == rhs_ ] :=
  lhs == Hff[h,g] rhs
EnhCoup[other_] = other

M$CouplingMatrices =
  EnhCoup/@ M$CouplingMatrices

Note that the enhancement factors depend on the fermion generation. This is because the couplings are defined at the classes level, i.e. for the class F[4] of down-type quarks {d, s, b}, not for the bottom quark alone. Thus one needs to set

Hff[h,1] = Hff[h,2] = 1.

A printout of the new Feynman rules can be obtained with the WriteTeXFile.m program that comes with FeynArts.

4. Scripting Mathematica

Question: How can I do efficient batch processing with Mathematica?

Answer: Put everything into a script, using sh’s Here documents:

#! /bin/sh ......... Shell Magic
math << \_EOF_ ..... start Here document
  << FeynArts‘
  << FormCalc‘
  top = CreateTopologies[...];
  ...
_EOF_ .............. end Here document

Everything between “<< \tag” and “tag” goes to Mathematica as if it were typed from the keyboard. Note the “\” before tag, it makes the shell pass everything literally to Mathematica, without shell substitutions. This is important because Mathematica uses many characters which have a special meaning to the shell, such as $, [, {, etc.

The advantages of this method are:

1. Everything is contained in one compact shell script (i.e., a text file), even if it involves several Mathematica sessions.

2. Such a script can easily be run in the background, or combined with utilities such as make.

3. One can seamlessly combine Mathematica and shell programming. A slightly subtle issue is how to get shell variables such as command-line arguments into Mathematica even though substitutions in the Here file are turned off through the use of \ in \tag. Solution: pass them on the command-line when invoking Mathematica:

#! /bin/sh
math -run "arg1=$1" ... << \END
  ...
END

Debugging hint: the -x flag makes sh echo every statement, it can be added after the Shell Magic:

#! /bin/sh -x

5. Not the Cross-Section

Question: Can I get things out of FormCalc other than the cross-section? Can I, for example, compute the Wilson coefficients for b → sγ?

Answer: The relevant operators for b → sγ, the prefactors of which are the Wilson coefficients, are

O_7 = e 16π^2 m_b ⟨s_j | ω_+σ_{μν} |b_i⟩ δ_{ij} F^{μν} (1) O_8 = g_s 16π^2 m_b ⟨s_j | ω_+σ_{μν} |b_i⟩ T^a_{ij} G^{μν}_a (2) Generating the partonic diagrams and amplitudes is of course a standard exercise:

tops = CreateTopologies[1, 1 -> 2]
ins = InsertFields[tops,
  F[4,{3}] -> {F[4,{2}], V[1]} ]
amp = CalcFeynAmp[CreateFeynAmp[ins],
  FermionChains -> Chiral ]

For the O_8 coefficient, the photon V[1] has to be replaced by the gluon V[5]. The FermionChains option ensures that the result contains the chiral Dirac chains needed to read off the Wilson coefficients, rather than the default Weyl chains.

CalcFeynAmp collects Dirac chains and colour

matrices in a function Mat. For identifying the operators it is thus simplest to construct a function id that replaces Mat.

The identification of the operators is slightly nontrivial because FormCalc strives to reduce the number of Lorentz indices as much as possible and thus turns ω_±γ^μγ^ν into ω_±γ^μ. This can be reversed by applying the Gordon identity

⟨s_2| ω_±γ^μ |s_1⟩ =

(p_1 + p_2)^μ m_1

⟨s_2| ω_± |s_1⟩ +

i(p_2 − p_1)_ν

m_1

⟨s_2| ω_±σ^{μν} |s_1⟩ . (3)

Since this is contracted with ε_μ, the photon/gluon polarization vector, we can use the momentumspace correspondence σ^{μν} F_{μν} → 2σ^{μν}(p_2−p_1)_ν ε_μ to identify the F_{μν} -term, thus

⟨s_2| ω_±/ε |s_1⟩ = 2 m_1 ε · p_1 ⟨s_2| ω_± |s_1⟩ + i 2m_1 ⟨s_2| ω_±σ^{μν} |s_1⟩ F_{μν} . (4)

id[r_. DiracChain[s2_Spinor, om_, eps_,
           s1:Spinor[p1_, m1_, _]]] :=
  2/m1 r Pair[eps, p1] *
    DiracChain[s2, om, s1] +
  I/(2 m1) id[r sig[om]]

It is not even necessary to explicitly write out the σ^{μν} -term since its only purpose is to match the operators:

id[r_. sig[om_] SUNT[i_, j_]] :=
  r O7[om]/(EL MB/(16 Pi^2))
id[r_. sig[om_] SUNT[a_, i_, j_]] :=
  r O8[om]/(GS MB/(16 Pi^2))

Now we can apply this function to the amplitude:

amp = Plus@@ amp //. Abbr[] /. Mat -> id
c7 = Coefficient[amp, O7[6]]
c8 = Coefficient[amp, O8[6]]

Using FormCalc’s output functions it is also pretty straightforward to turn these expressions into Fortran code:

file = OpenFortran["bsgamma.F"]
WriteString[file,
  SubroutineDecl["bsgamma(C7, C8)"] <>
  "\tdouble complex C7, C8\n" <>
  "#include \"model.h\"\n" <>
  "#include \"looptools.h\"\n"]
WriteExpr[file, {C7 -> c7, C8 -> c8}]
WriteString[file, "\tend\n"]
Close[file]

6. Abbreviations

The automated introduction of abbreviations is one of the key concepts in FormCalc. It is crucial in rendering an amplitude as compact as possible. The main effect comes from three layers of recursively defined abbreviations, introduced when the amplitude is read back from FORM, i.e. during

CalcFeynAmp. For example:

 AbbSum29 = Abb2 + Abb22 + Abb23 + Abb3

         Abb22 = Pair1 Pair3 Pair6

              Pair3 = Pair[e[3], k[1]]

Written out, this abbreviation is equivalent to

Pair[e[1],e[2]] Pair[e[3],k[1]] Pair[e[4],k[1]] +
Pair[e[1],e[2]] Pair[e[3],k[2]] Pair[e[4],k[1]] +
Pair[e[1],e[2]] Pair[e[3],k[1]] Pair[e[4],k[2]] +
Pair[e[1],e[2]] Pair[e[3],k[2]] Pair[e[4],k[2]]

In addition to these abbreviations assigned by

CalcFeynAmp, FormCalc introduces another set of

abbreviations for the loop integrals when generating Fortran code, i.e. during WriteSquaredME.

6.1. Categories

Both of the aforementioned types of abbreviations, but in particular the latter, are costly in CPU time. It is thus key to a decent performance that the abbreviations are grouped into different categories:

1. Abbreviations that depend on the helicities.

2. Abbreviations that depend on angular variables.

3. Abbreviations that depend only on √ s. Correct execution of the different categories guarantees that almost no redundant evaluations are made. For example, for a 2 → 2 process with external unpolarized fermions, statements in the innermost loop over the helicities are executed 2^4 times as often as those in the loop over the angle. This technique of moving invariant expressions out of the loop is known as ‘hoisting’ in computer science.

6.2. Common Subexpressions

Another optimization method, common subexpression elimination, is implemented in the function OptimizeAbbr and can yield some additional

10–30% speed-up. It works in two steps. First, redundant parts are removed, e.g. the abbreviations

AbbSum2 -> Abb9 - Abb87
AbbSum8 -> Abb9 - Abb13 - Abb74 - Abb87
AbbSum9 -> Abb9 + Abb13 + Abb74 - Abb87

are replaced by

AbbSum2 -> Abb9 - Abb87
AbbSum8 -> AbbSum2 - Abb13 - Abb74
AbbSum9 -> AbbSum2 + Abb13 + Abb74

In the second step, common parts are put into temporary variables, thereby simplifying the last lines further to

AbbSum2 -> Abb9 - Abb87
help1 -> Abb13 + Abb74
AbbSum8 -> AbbSum2 - help1
AbbSum9 -> AbbSum2 + help1

Both optimization techniques together make the generated code essentially as fast as hand-tuned code.

6.3. The Abbreviate Function

The new Abbreviate function extends the advantages of the abbreviation system to arbitrary expressions. Its usage is for example:

abbrexpr = Abbreviate[expr, 5]

The second argument, 5, determines the level below which abbreviations are introduced. The level determines how much of expression is ‘abbreviated away,’ i.e. how much of the structure is preserved. In the extreme, for a level of 1, the result is just a single symbol. Abbreviationing also has the ‘side effect’ that duplicate expressions are replaced by the same symbol.

This new type of abbreviations for subexpressions has to be retrieved separately from the other ones with Subexpr[].

The most important option of Abbreviate is

Preprocess. It is used e.g. as follows:

abbrexpr = Abbreviate[ expr, 5,
  Preprocess -> Simplify ] and specifies a function which is applied to each subexpression before the abbreviations are introduced.

At some 30 sec. execution time for Abbreviate, the typical speed-up was a factor 3 in MSSM calculations.

6.4. Auxiliary functions

When simplifying large expressions, it is often desirable to have a replacement for Simplify which is faster by performing only specific simplifications. This is because Simplify is quite efficient on short expressions but increasingly slow on longer ones.

OnSize constructs a special function for simpli-

fication that does different things depending on the size of its argument. For example,

f = OnSize[100, Simplify,
           500, DenCollect,
           Pool]

If the LeafCount of the argument of f is • below 100, Simplify is used, • between 100 and 500, DenCollect is used, • above 500, Pool is used.

DenCollect collects terms with denominators

that are identical up to a numerical constant.

Pool combines terms with common factors.

Unlike Factor, it looks at the terms pairwise and can thus do ab + ac + d → a(b + c) + d fast. Pool will not factor out very small expressions because the effect of this on the size of the whole expression is typically not worth the effort.

7. Summary

We presented several examples of programming techniques which enable the user to extend the capabilities of FeynArts and FormCalc with comparatively little effort:

• programming diagram filters,

• applying (small) changes to model files,

• scripting Mathematica,

• computing quantities other than the crosssection.

In addition, we elucidated the important concept of abbreviations in FormCalc which is extended to arbitrary expressions by the new Abbreviate function.

Acknowledgements

TH thanks Granada University for kind hospitality during part of this work.

REFERENCES

1. T. Hahn, Comp. Phys. Commun. 140 (2001) 418 [hep-ph/0012260].

2. T. Hahn, M. Pérez-Victoria, Comp. Phys.

Commun. 118 (1999) 153 [hep-ph/9807565]. 3. T. Hahn, Nucl. Phys. Proc. Suppl. 89 (2000) 231 [hep-ph/0005029].